Double somme de Riemann

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2006. L'objectif principal de cette feuille Maple est la présentation de l'intégrale double à l'aide de la double somme de Riemann. Après un rappel de l'intégrale définie d'une fonction d'une variable à l'aide d'une partition régulière, l'élève réalisera la similitude de la démarche avec les sommes de Riemann dans la définition de l'intégrale double.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
> with(plots,coordplot,display,setoptions,setoptions3d,textplot,
  pointplot3d):
  setoptions(axesfont=[TIMES,ROMAN,8]);
  setoptions3d(axesfont=[TIMES,ROMAN,8],lightmodel=light1,axes=framed,
  labels=[x,y,``]);
> with(plottools,cuboid,rectangle,transform):
```

Trois procédures à initialiser

Petit rappel

Rappelons d'abord la définition de l'intégrale définie au sens de Riemann d'une fonction réelle d'une seule variable réelle. Soit f une fonction définie sur un intervalle $[a, b]$.
Divisons l'intervalle $[a, b]$ en n sous-intervalles. Soit $\boldsymbol{P} = \{x_0, x_1, x_2, x_3, \ldots, x_{n-1}, x_n\}$ une partition de l'intervalle $[a, b]$ où

$$x_0 = a < x_1 < x_2 < x_3 < \ldots < x_{i-1} < x_i < x_{i+1} < \ldots < x_{n-1} < x_n = b.$$

La partition $\boldsymbol{P}$ détermine donc n sous-intervalles $[{}_{i-1}, x_i]$ telle que $x_0 = a$ et $x_n = b$.
Soit $c_i \in [x_{i-1}, x_i]$ un représentant arbitraire du $i^{ième}$ sous-intervalle de l'intervalle $[a, b]$
Soit $\Delta x_i = |x_i - x_{i-1}|$, la largeur du $i^{ième}$ sous-intervalle $[x_{i-1}, x_i]$.

On définit, sur l'intervalle $[a, b]$, une somme intégrale (somme de Riemann) comme étant la somme

$$SR_{[a,b]} = \sum_{i=1}^{n} \mathrm{f}(c_i)\, \Delta x_i. \quad \text{où} \quad c_i \in [x_{i-1}, x_i]$$

D'une part, la définition de $SR_{[a,b]}$ n'implique pas que la partition $\boldsymbol{P}$ soit régulière, c'est-à-dire que pour tout indice i, nous n'avons pas nécessairement

$\Delta x_i = \dfrac{b-a}{n}$. D'autre part, on peut créer, avec une même partition ***P***, autant de sommes intégrales différentes qu'il y a de manières de choisir le représentant c_i dans chaque sous-intervalle $[\,x_{i-1}, x_i\,]$.

Lorsque la limite $\lim_{\max \Delta x_i \to 0} \left(\sum_{i=1}^{n} f(c_i)\, \Delta x_i \right)$ **existe** pour une fonction f définie sur un intervalle $[a, b]$, nous disons que la fonction f est intégrable au sens de Riemann et que la valeur de cette limite est appelée l'intégrale définie de la fonction f sur [a,b] et sera notée $\int_a^b f(x)\, dx$.

$$\int_a^b f(x)\, dx = \lim_{n \to \infty,\ \max \Delta x_i \to 0} \left(\sum_{i=1}^{n} f(c_i)\, \Delta x_i \right), \qquad c_i \in [x_{i-1}, x_i]$$

Il est prouvé que si la limite $\lim_{\max \Delta x_i \to 0} \left(\sum_{i=1}^{n} f(c_i)\, \Delta x_i \right)$ existe, elle est indépendante du représentant choisi c_i.

Dans le cas où les partitions successives sont régulières, pour tout i nous avons $\Delta x_i = \Delta x = \dfrac{b-a}{n}$ et ainsi la définition est

$$\int_a^b f(x)\, dx = \lim_{n \to \infty} \left(\sum_{i=1}^{n} f(c_i)\, \Delta x \right), \qquad c_i \in [x_{i-1}, x_i]$$

Dans le cas où la fonction f est **continue,** la fonction f est intégrable. De plus, si elle est **non négative** sur $[a, b]$, $\int_a^b f(x)\, dx$ représente l'aire de la surface bornée par le tracé de f et l'axe des x entre $x = a$ et $x = b$.

Par exemple, considérons la fonction f définie par $f(x) = 2 + x^2$ sur l'intervalle $[1, 3]$.

```
> f:=x->2+x^2;
```

$$f := x \mapsto 2 + x^2 \tag{2.1}$$

Illustrons, avec cette fonction, une somme droite de Riemann $SR_{[1,3]}$ en subdivisant l'intervalle $[1, 3]$ en 32 sous-intervalles.

```
> Courbe:=plot([x,f(x),x=-1..4],thickness=2,color="Niagara 12"):
  Animation:=Student[Calculus1][ApproximateInt](f(x), 1 .. 3,
  method=midpoint,output = animation, partition=1,refinement=halve,
  subpartition=all, iterations = 6, showpoints = true, boxoptions =
  [filled = [color = "Niagara 10", transparency = .5]]):
> display([Courbe,Animation],view=[-0.5..4.,0..14]);
```



Une approximation animée de $\int_1^3 f(x)\,dx$ par une somme supérieure de Riemann, où $f(x)$

Cliquez sur le graphique précédent avec un clic gauche puis lancer l'animation en cliquant ensuite dans la barre de menu contextuel le bouton de lancement (le bouton play ▶).

Développons la limite à l'infini de la somme de Riemann $SR_{[1,3]}$ en choisissant systématiquement le représentant comme le point milieu de chaque sous-intervalle $[x_{i-1}, x_i]$.

```
> n:='n':
> SR:=Student[Calculus1][ApproximateInt](f(x), x = 1 .. 3, method=
  midpoint,partition=n,output = sum);
```

$$SR := \frac{2\left(\sum_{i=0}^{n-1}\left(\left(1+\frac{2\left(i+\frac{1}{2}\right)}{n}\right)^2+2\right)\right)}{n} \tag{2.2}$$

```
> Int(f(x),x=1..3)=Limit(SR,n=infinity);
  ``=Limit(expand(value(SR)),n=infinity);
  ``=value(rhs(%));
  evalf(%);
```

$$\int_1^3 \left(x^2+2\right)\,dx = \lim_{n\to\infty} \frac{2\left(\sum_{i=0}^{n-1}\left(\left(1+\frac{2\left(i+\frac{1}{2}\right)}{n}\right)^2+2\right)\right)}{n}$$

$$= \lim_{n\to\infty}\left(\frac{38}{3}-\frac{2}{3n^2}\right)$$

$$= \frac{38}{3}$$

$$= 12.6666666700 \tag{2.3}$$

Le tableau ci-dessous nous donne différentes approximations de l'aire avec des partitions régulières de l'intervalle [1,3] de plus en plus fines avec des représentants choisis à gauche, au milieu et à droite de chaque sous-intervalles de partitions régulières.

```
> printf(`\n                    Sommes intégrales avec partitions
  régulières pour f\n`);
  printf(`\n    |        n      |      SR[gauche]     |       SR[médian]
    |     SR[droit]     |\n`);
  printf(`   ==========================================================
  ======================\n`);
   for k in [$(2^i, i =0 .. 16)]
   do
   printf(`   |    %5d      |  %15.10f   |   %15.10f    | %15.10f
  |\n`,
   k,
   Approxint1d(f(x),x=1..3,k,gauche),
   Approxint1d(f(x),x=1..3,k,médian),
   Approxint1d(f(x),x=1..3,k,droit))
  od;
  k:='k':
```

```
                 Sommes intégrales avec partitions régulières pour f

   |        n      |      SR[gauche]     |       SR[médian]      |      SR[droit]      |
   ================================================================================
   |        1      |     6.0000000000    |     12.0000000000     |    22.0000000000    |
   |        2      |     9.0000000000    |     12.5000000000     |    17.0000000000    |
   |        4      |    10.7500000000    |     12.6250000000     |    14.7500000000    |
   |        8      |    11.6875000000    |     12.6562500000     |    13.6875000000    |
   |       16      |    12.1718750000    |     12.6640625000     |    13.1718750000    |
   |       32      |    12.4179687500    |     12.6660156250     |    12.9179687500    |
   |       64      |    12.5419921875    |     12.6665039062     |    12.7919921875    |
   |      128      |    12.6042480469    |     12.6666259766     |    12.7292480469    |
   |      256      |    12.6354370117    |     12.6666564941     |    12.6979370117    |
   |      512      |    12.6510467529    |     12.6666641235     |    12.6822967529    |
   |     1024      |    12.6588554382    |     12.6666660309     |    12.6744804382    |
   |     2048      |    12.6627607346    |     12.6666665077     |    12.6705732346    |
   |     4096      |    12.6647136211    |     12.6666666269     |    12.6686198711    |
   |     8192      |    12.6656901240    |     12.6666666567     |    12.6676432490    |
   |    16384      |    12.6661783904    |     12.6666666642     |    12.6671549529    |
   |    32768      |    12.6664225273    |     12.6666666660     |    12.6669108085    |
   |    65536      |    12.6665445967    |     12.6666666665     |    12.6667887373    |
```

Le comportement numérique médian est plutôt éloquent quant à la convergence de la somme de Riemann.

Pour terminer, évaluons directement $\int_1^3 (2+x^2)\,\mathrm{d}x$.

```
> Int(f(x),x=1..3)=int(f(x),x=1..3);
  ``=evalf[15](rhs(%));
  convert(%,rational);
```

$$\int_1^3 (x^2+2)\,\mathrm{d}x = \frac{38}{3}$$

$$= 12.6666666667$$

$$= \frac{38}{3} \quad \textbf{(2.4)}$$

Double somme de Riemann et intégrale double

Nous allons voir que la similitude entre de l'intégrale double et l'intégrale simple est assez remarquable. Nous allons introduire une somme de Riemann comme calcul d'un volume semblablement à ce que l'on fait pour une somme de Riemann en tant que calcul d'aire dans le cas d'une fonction positive d'une seule variable.



Requêtes pour la figure précédente

Soit une fonction réelle f de deux variables réelles définie sur un ensemble D $\subseteq \mathbb{R}^2$.

Considérons un pavé R = [a,b] × [c,d] = $\{ (x, y) \mid a \le x \le b,\ c \le y \le d \} \subseteq$ D.



Requêtes pour la figure précédente

La première étape consiste à diviser le rectangle R en sous-rectangles. Divisons l'intervalle $[a, b]$ en m sous-intervalles $[x_{i-1}, x_i]$ d'égales longueurs $\Delta x = \dfrac{b-a}{m}$ et l'intervalle $[c, d]$ en n sous-intervalles $[y_{i-1}, y_i]$ d'égales longueurs $\Delta y = \dfrac{d-c}{n}$.



Requêtes pour la figure précédente

Chaque sous-rectangle $R_{ij} = \left[x_{i-1}, x_i\right] \times \left[y_{j-1}, y_i\right] = \{ (x, y) \,|x_{i-1} \le x \le x_i,\ y_{j-1} \le y \le y_i \}$ a une aire $\Delta A = \Delta x \cdot \Delta y$.

Choisissons un représentant $(\alpha_{ij}, \beta_{ij})$ quelconque du sous-rectangle R_{ij} où

$$x_{i-1} \le \alpha_{ij} \le x_i,\ y_{j-1} \le \beta_{ij} \le y_i$$

et considérons le produit

$$\mathrm{f}\left(\alpha_{ij}, \beta_{ij}\right) \Delta A$$

Ce produit correspond au volume du R_{ij}*-ième* parallélépipède droit. La valeur de ce volume n'est pas loin d'avoir la même valeur que la portion de volume comprise sous la surface et au-dessus du rectangle R_{ij}.



Requêtes pour la figure précédente

Faisons ensuite une première somme de Riemann de toutes les colonnes en sommant en y: $\sum_{j=1}^{n} f\left(\alpha_{ij}, \beta_{ij}\right) \Delta A$



Requêtes pour la figure précédente

Faisons ensuite une seconde somme de Riemann toutes les colonnes en sommant en x:

$$\sum_{i=1}^{m}\left(\sum_{j=1}^{n} f(\alpha_{ij}, \beta_{ij})\,\Delta A\right)$$



Requêtes pour la figure précédente

L'intégrale double de f sur le rectangle R (où chaque partition est régulière)

$$\iint_R f(x,y)\,\mathrm{d}A = \lim_{m,n\to\infty}\left(\sum_{i=1}^{m}\left(\sum_{j=1}^{n} f(\alpha_{ij}, \beta_{ij})\,\Delta A\right)\right)$$

pourvu que cette limite existe.

Il est démontré que la limite dont il est précédemment question existe pour toutes les fonctions continues.

Dans une double somme de Riemann, le représentant $(\alpha_{ij}, \beta_{ij})$ peut être n'importe quel point du rectangle $R_{i,j}$, mais cela simplifie les calculs à faire, si nous choisissons le représentant dans le coin supérieur droit, dans le coin inférieur gauche ou dans le milieu du rectangle.

Considérons une fonction f non négative sur D. Par exemple, soit la fonction f définie par $\mathrm{f}(x,y) = 16 - x^2 - 2y^2$ sur $R = [0,2] \times [0,2]$. Évaluons approximativement le volume du solide qui repose sur le carré R fermé par la surface paraboloïque elliptique f.

```
> f:=(x,y)->16-x^2-2*y^2:
'f'(x,y)=f(x,y);
Surface:=plot3d([x,y,f(x,y)],x=0..2,y=0..2,filled=true,color="Niagara 4",style=patchnogrid,grid=[20,20],
orientation=[25,75],transparency=0.45);
```

$$f(x,y) = -x^2 - 2y^2 + 16$$



Divisons le carré R en quatre carrés égaux. Chaque carré nous permet de poser $\Delta A = 1$. En choisisssant le coin supérieur droit dans chaque sous-rectangle comme représentant, $(1,1), (1,2), (2,1)$ et $(2,2)$, nous pouvons poser

```
> unassign('i,j');
  V_approx=Sum(Sum('f'(i,j)*`&Delta;A`,j=1..2),i=1..2);
```

$$V\_approx = \sum_{i=1}^{2} \sum_{j=1}^{2} f(i,j)\, \Delta A \quad \textbf{(3.1)}$$

```
> value(subs(`&Delta;A`=1,(3.1)));
```

$$V\_approx = 34 \quad \textbf{(3.2)}$$

```
> Surface:=plot3d([x,y,f(x,y)],x=0..2,y=0..2,style=patchnogrid,color=
  "Niagara 4",transparency=0.25):
  Colonne:=Colonnes(f(x,y),x=0..2,y=0..2,2,2,droit):
  Points:=pointplot3d({[1,1,f(1,1)],[1,2,f(1,2)],[2,1,f(2,1)],[2,2,f(2,
  2)]},color="Niagara 4",symbol=solidsphere,symbolsize=12):
  display(Points,Colonne,Surface,lightmodel=light4,orientation=[25,75],
          title="Approximation du volume sous la surface");
```



Cette valeur approximative peut être améliorée si nous augmentons de nombre ce carrés.

```
> for i from 2 to 18 do
   Blocs||(i):=display(Colonnes(f(x,y),x=0..2,y=0..2,i,i,droit)):
   Tracé||(i):=display([Blocs||(i)],Surface):
  od:
```

```
> display([Tracé||(2..18)],insequence=true,lightmodel=light4,
  orientation=[25,75],
          title="Approximation du volume sous la surface");
```

Approximation du volume sous la surface



Cliquez sur le graphique précédent avec un clic gauche puis lancer l'animation en cliquant ensuite dans la barre de menu contextuel le bouton de lancement (le bouton play ►).

Pour mieux opérer la double somme de Riemann avec des carrés de plus en plus fins, nous allons reformuler la double somme en prenant en charge la valeur de ΔA en tant que produit $\Delta x\,\Delta y$.

```
> unassign('a,b,i,j,dx,dy,`&Delta;x`,`&Delta;y`');
  dx:=`&Delta;x`:
  dy:=`&Delta;y`:
```

```
> Sum(Sum('f'(a+i*dx,b+j*dy)*dx,i=1..m),j=1..n)*dy;
```

$$\left(\sum_{j=1}^{n}\ \sum_{i=1}^{m} f\left(i\,\Delta x + a, j\,\Delta y + b\right)\,\Delta x\right)\Delta y \qquad \textbf{(3.3)}$$

Nous aurions pu tout aussi reformuler la double somme en prenant en charge la valeur de ΔA en tant que produit $\Delta y\,\Delta x$

```
> Sum(Sum('f'(a+i*dx,b+j*dy)*dy,i=1..m),j=1..n)*dx;
```

$$\left(\sum_{j=1}^{n}\ \sum_{i=1}^{m} f\left(i\,\Delta x + a, j\,\Delta y + b\right)\,\Delta y\right)\Delta x \qquad \textbf{(3.4)}$$

Ainsi, recalculons la première approximation.

```
> dx := (2-0)/2:
```

```
dy := (2-0)/2:
Sum(Sum(subs(x=0+i*dx,y=0+j*dy,'f(x,y)'),i=1..2),j=1..2)*dx*dy=sum
(sum(subs(x=0+'i'*dx,y=0+'j'*dy,f(x,y)),'i'=1..2),'j'=1..2)*dx*dy;
```

$$\sum_{j=1}^{2} \sum_{i=1}^{2} f(i,j) = 34 \tag{3.5}$$

Calculons la double intégrale en calculant la limite à l'infini d'une double somme de Riemann (avec des partitions régulières).

```
> a:=0:
  b:=2:
  c:=0:
  d:=2:
```

```
> dx := (b-a)/m:
  dy := (d-c)/n:
  V=Limit(Limit(Sum(Sum(subs(x=a+i*dx,y=c+j*dy,'f(x,y)'),i=1..m),j=1..
  n)*dx*dy,n=infinity),m=infinity);
  ``=Limit(Limit(sum(sum(subs(x=a+i*dx,y=c+j*dy,f(x,y)),i=1..m),j=1..n)
  *dx*dy,n=infinity),m=infinity);
```

$$V = \lim_{m \to \infty} \lim_{n \to \infty} \frac{4 \left(\sum_{j=1}^{n} \sum_{i=1}^{m} f\left(\frac{2\,i}{m}, \frac{2\,j}{n} \right) \right)}{m\,n}$$

$$= \lim_{m \to \infty} \lim_{n \to \infty} \frac{1}{m\,n} \left(4 \left(\frac{44\,(n+1)\,m}{3} - 2\,n - \frac{2\,(n+1)}{3\,m} - \frac{8\,m\,(n+1)^3}{3\,n^2} + \frac{4\,m\,(n+1)^2}{n^2} - \frac{4\,m\,(n+1)}{3\,n^2} - \frac{44\,m}{3} + \frac{2}{3\,m} \right) \right) \tag{3.6}$$

```
> expand(%);
```

$$= \lim_{m \to \infty} \frac{8 \left(\lim_{n \to \infty} \left(18\,m^2 - \frac{6\,m^2}{n} - 3\,m - \frac{2\,m^2}{n^2} - 1 \right) \right)}{3\,m^2} \tag{3.7}$$

```
> value((3.7));
```

$$= 48 \tag{3.8}$$

Le tableau ci-dessous nous donne différentes approximations du volume avec des partitions régulières de plus en plus fines.

```
> printf(`\n          Doubles sommes intégrales avec partitions
  régulières pour f(x,y)\n`);
  printf(`\n   |        n     |      SR[gauche]    |       SR[médian]
    |     SR[droit]     |\n`);
  printf(`   ==========================================================
  ======================\n`);
   for k in [1,2,4,8,16,32,64,128,256,512,1024,2048,4096,8192,16384]
   do
```

```
 printf(`   |    %5d     |  %15.10f   |   %15.10f    | %15.10f
|\n`,
 k,
 Approxint2d(f(x,y),x=0..2, y=0..2 ,k ,k,gauche),
 Approxint2d(f(x,y),x=0..2, y=0..2 ,k ,k,médian),
 Approxint2d(f(x,y),x=0..2, y=0..2 ,k ,k,droit))
od;
k:='k':
```

```
      Doubles sommes intégrales avec partitions régulières pour f(x,y)

|        n      |      SR[gauche]     |        SR[médian]       |      SR[droit]      |
==================================================================================
|        1      |    64.0000000000    |      52.0000000000      |    16.0000000000    |
|        2      |    58.0000000000    |      49.0000000000      |    34.0000000000    |
|        4      |    53.5000000000    |      48.2500000000      |    41.5000000000    |
|        8      |    50.8750000000    |      48.0625000000      |    44.8750000000    |
|       16      |    49.4687500000    |      48.0156250000      |    46.4687500000    |
|       32      |    48.7421875000    |      48.0039062500      |    47.2421875000    |
|       64      |    48.3730468750    |      48.0009765625      |    47.6230468750    |
|      128      |    48.1870117188    |      48.0002441406      |    47.8120117188    |
|      256      |    48.0936279297    |      48.0000610352      |    47.9061279297    |
|      512      |    48.0468444824    |      48.0000152588      |    47.9530944824    |
|     1024      |    48.0234298706    |      48.0000038147      |    47.9765548706    |
|     2048      |    48.0117168427    |      48.0000009537      |    47.9882793427    |
|     4096      |    48.0058588982    |      48.0000002384      |    47.9941401482    |
|     8192      |    48.0029295683    |      48.0000000596      |    47.9970701933    |
|    16384      |    48.0014648139    |      48.0000000149      |    47.9985351264    |
```

Le comportement numérique médian est plutôt éloquent quant à la convergence de la double somme de Riemann.

Dans le cas où la formulation de la double somme de Riemann prend en charge la valeur de ΔA en tant que produit $\Delta x\,\Delta y$, nous disons que nous sommons d'abord par rapport à la variable x.

Dans le cas où la formulation de la double somme de Riemann prend en charge la valeur de ΔA en tant que produit $\Delta y\,\Delta x$, nous disons que nous sommons d'abord par rapport à la variable y.

```
> unassign(i,j);
  t:=transform((x,y)->[x,y,0]):
  A:=coordplot(cartesian,[0..2,0..2],grid=[21,21],linestyle=[1,1],
  color=["Niagara 4","Niagara 4"]):
  Surface:=plot3d([x,y,f(x,y)],x=0..2,y=0..2,filled=true,color="Niagara
  4",style=patchnogrid,grid=[20,20],transparency=0.65):
  Colonneyx:= n->display(Surface,t(A),Colonnes(f(x,y),x=(k-1)*0.1..k*
  0.1,y=0..2,1,20,médian),Surface):
  Colonnexy:= n->display(Surface,t(A),Colonnes(f(x,y),x=0..2,y=(k-1)*
  0.1..k*0.1,20,1,médian),Surface):
  Opts_affichage:=tickmarks=[[0='a',0.9=x[i-1],1=x[i],2='b'],\
                  [0='c',0.9=y[j-1],1=y[j],2='d'], 0],\
```

```
                    axesfont=[TIMES,ROMAN,10],\
                    orientation=[25,60]:
XY:=display(seq(Colonnexy(k),k=1..20),insequence=true,Opts_affichage)
:
YX:=display(seq(Colonneyx(k),k=1..20),insequence=true,Opts_affichage)
:
> display(Matrix(1,2,[XY,YX]));
```



Pour terminer, évaluons directement les intégrales doubles $\int_0^2 \int_0^2 f(x,y)\, dx\, dy$ et $\int_0^2 \int_0^2 f(x,y)\, dy\, dx$ qui seront obtenues comme des intégrales itérées.

```
> Int(Int(f(x,y),x=0..2),y=0..2)=int(int(f(x,y),x=0..2),y=0..2);
```

$$\int_0^2 \int_0^2 (-x^2 - 2y^2 + 16)\, dx\, dy = 48 \tag{3.9}$$

```
> Int(Int(f(x,y),y=0..2),x=0..2)=int(int(f(x,y),y=0..2),x=0..2);
```

$$\int_0^2 \int_0^2 (-x^2 - 2y^2 + 16)\, dy\, dx = 48 \tag{3.10}$$

Théorème de Fubini

Il y a égalité selon les deux ordres d'intégration à la condition que la fonction f soit continue sur un rectangle $R = \{(x,y) | a \le x \le b,\ c \le y \le d\}$. Plus généralement, il y a égalité seulement lorsque f est bornée sur R, seulement si f est discontinue en un nombre fini de courbe lisses et que les intégrales itérées existes.